set matsize 11000

* ) Insert directory
cd ""
use "Author-paper-referee level data", clear

***************************************
global authorcontrols female gender_missing published_papers published_from_5_no_pp USNEWS_1-USNEWS_5 is_nber ///
	employmentrank1 employmentrank2 employmentrank3 employmentrank4 employmentrank5 NBER_AG-NBER_PE degree_1 degree_2

sum $authorcontrols

* ) Top-coding distance and include missing
replace referee_distance = 6 if referee_distance>6 | referee_distance==.
replace editor_distance = 6 if editor_distance>6 | editor_distance==.

*) When doing editor analysis, don't double count observations due to referee assignment
bysort ms_number_id editor_id author_id: gen indicate = 1 if _n==1

* ) For model with "number of direct connections between author and reviewer/editor"
gen referee_one_degree = (referee_distance==1)
gen editor_one_degree = (editor_distance==1)
gen referee_Nconnections = referee_one_degree + referee_phdmatch + referee_employmentmatch + referee_isnbermatch_program 
gen editor_Nconnections = editor_one_degree + editor_phdmatch + editor_employmentmatch + editor_isnbermatch_program 
drop referee_one_degree editor_one_degree
*Generating indicators for each category
tab referee_Nconnections, generate(referee_Nconnections_raw) m
tab editor_Nconnections, generate(editor_Nconnections_raw) m
*Top code 2+ connections
replace referee_Nconnections_raw3 = referee_Nconnections_raw3+referee_Nconnections_raw4
replace editor_Nconnections_raw3 = editor_Nconnections_raw3+editor_Nconnections_raw4+editor_Nconnections_raw5

**************
* ) Outcome variable - Any acceptance
gen accept = (status=="Accepted")

*****************************************
*****************************************
*****************************************
* ) Conditional on getting past the desk and RR
eststo clear

bysort ms_number_id author_id: egen RRavg = mean(RR)
tab RRavg if passed_the_desk==1, m

*PhD Institution 
eststo: reghdfe accept RRavg editor_phdmatch $authorcontrols [aweight=editor_weight] if indicate==1 & passed_the_desk==1, absorb(editor_id) vce(cluster ms_number_id)
estadd local refFE "X",replace
estadd local paperFE " ",replace
estadd local authorcontrols "X",replace
*Same Employment
eststo: reghdfe accept RRavg editor_employmentmatch $authorcontrols [aweight=editor_weight] if indicate==1 & passed_the_desk==1, absorb(editor_id) vce(cluster ms_number_id)
estadd local refFE "X",replace
estadd local paperFE " ",replace
estadd local authorcontrols "X",replace
*NBER programs
eststo: reghdfe accept RRavg editor_isnbermatch_program $authorcontrols [aweight=editor_weight] if indicate==1 & passed_the_desk==1, absorb(editor_id) vce(cluster ms_number_id)
estadd local refFE "X",replace
estadd local paperFE " ",replace
estadd local authorcontrols "X",replace
*Degrees of separation
eststo: reghdfe accept RRavg editor_distance_raw1-editor_distance_raw3 $authorcontrols [aweight=editor_weight] if indicate==1 & passed_the_desk==1, absorb(editor_id) vce(cluster ms_number_id)
estadd local refFE "X",replace
estadd local paperFE " ",replace
estadd local authorcontrols "X",replace
*Everything together
eststo: reghdfe accept RRavg editor_phdmatch editor_employmentmatch editor_isnbermatch_program editor_distance_raw1-editor_distance_raw3 $authorcontrols [aweight=editor_weight] if indicate==1 & passed_the_desk==1, absorb(editor_id) vce(cluster ms_number_id)
estadd local refFE "X",replace
estadd local paperFE " ",replace
estadd local authorcontrols "X",replace
*Number of connections
eststo: reghdfe accept RRavg editor_Nconnections_raw2-editor_Nconnections_raw3 $authorcontrols [aweight=editor_weight] if indicate==1 & passed_the_desk==1, absorb(editor_id) vce(cluster ms_number_id)
estadd local refFE "X",replace
estadd local paperFE " ",replace
estadd local authorcontrols "X",replace

label var RRavg "\underline{Outcome: Published} \\ \ \ Fraction positive review"
label var editor_phdmatch "\ \ Exact same PhD institution "
label var editor_employmentmatch "\ \ Former/current colleagues"
label var editor_isnbermatch_program "\ \ Same NBER program(s)"
label var editor_distance_raw1 "\ \ Degrees of separation: 1"
label var editor_distance_raw2 "\ \ Degrees of separation: 2"
label var editor_distance_raw3 "\ \ Degrees of separation: 3"
label var editor_distance_raw4 "\ \ Degrees of separation: 4"
label var editor_distance_raw5 "\ \ Degrees of separation: 5"
label var editor_distance_raw6 "\ \ Degrees of separation: 6+"
label var editor_Nconnections_raw2 "\hline \# of direct matches: \\ \ \ - One match"
label var editor_Nconnections_raw3 "\ \ - Two+ matches"

esttab using "editor_clubmatch_accept_RR.tex", ///
    replace se(3) b(3) star(* 0.10 ** 0.05 *** 0.01) noobs nonum nonotes nogaps ///
	s(N refFE authorcontrols r2, label("Author-editor-papers" "Editor FE" "Author controls" "R-squared") fmt(%12.0f %12.3f)) ///
	mtitles("(1)" "(2)" "(3)" "(4)" "(5)" "(6)" "(7)" "(8)" "(9)" "(10)" "(11)" "(12)" "(13)" "(14)" "(15)") label ///
	keep(RRavg editor_phdmatch editor_employmentmatch editor_isnbermatch_program editor_distance* editor_Nconnections_raw2 editor_Nconnections_raw3)
